Functional programming languages
Part IV: monadic transformations, monadic programming

Xavier Leroy

INRIA Paris-Rocquencourt

MPRI 2-4, 2015–2016
Monads in programming language theory

Monads are a technical device (inspired from category theory) with several uses in programming:

- To structure denotational semantics and make them easy to extend with new language features. (E. Moggi, 1989.) Not treated in this lecture.
- To factor out commonalities between many program transformations and between their proofs of correctness.
- As a powerful programming techniques in pure functional languages. (P. Wadler, 1992; the Haskell community).
Outline

1. Introduction to monads

2. The monadic translation
   - Definition
   - Correctness
   - Application to some monads

3. Monadic programming
   - More examples of monads
   - Monad transformers

4. Bonus track: applicative structures

5. Bonus track: comonads
Commonalities between program transformations

Consider the conversions to exception-returning style, state-passing style, and continuation-passing style. For constants, variables and $\lambda$-abstractions, we have:

\[
\begin{align*}
[N] &= V(N) & [N] &= \lambda s. (N, s) & [N] &= \lambda k. k \ N \\
[x] &= V(x) & [x] &= \lambda s. (x, s) & [x] &= \lambda k. k \ x \\
[\lambda x. a] &= V(\lambda x. [a]) & [\lambda x. a] &= \lambda s. (\lambda x. [a], s) & [\lambda x. a] &= \lambda k. k (\lambda x. [a])
\end{align*}
\]

in all three cases, we return (put in some appropriate wrapper) the values $N$ or $x$ or $\lambda x. [a]$. 
Commonalities between program transformations

For `let` bindings, we have:

\[
\begin{align*}
\llbracket \text{let } x = a \text{ in } b \rrbracket &= \text{match } \llbracket a \rrbracket \text{ with } E(x) \rightarrow \llbracket b \rrbracket | V(x) \rightarrow \llbracket b \rrbracket \\
\llbracket \text{let } x = a \text{ in } b \rrbracket &= \lambda s. \text{match } \llbracket a \rrbracket s \text{ with } (x, s') \rightarrow \llbracket b \rrbracket s' \\
\llbracket \text{let } x = a \text{ in } b \rrbracket &= \lambda k. \llbracket a \rrbracket (\lambda x. \llbracket b \rrbracket k)
\end{align*}
\]

In all three cases, we extract (one way or another) the value contained in the computation \(\llbracket a \rrbracket\), bind it to the variable \(x\), and proceed with the computation \(\llbracket b \rrbracket\).
Commonalities between program transformations

Concerning function applications:

\[
[a \ b] = \text{match } [a] \text{ with } \\
\quad | \ E(e_a) \rightarrow E(e_a) \\
\quad | \ V(v_a) \rightarrow \\
\quad \text{match } [b] \text{ with } E(e_b) \rightarrow E(e_b) | V(v_b) \rightarrow v_a \ v_b
\]

\[
[a \ b] = \lambda s. \text{match } [a] \ s \text{ with } (v_a, s') \rightarrow \\
\quad \text{match } [b] \ s' \text{ with } (v_b, s'') \rightarrow v_a \ v_b \ s''
\]

\[
[a \ b] = \lambda k. [a] (\lambda v_a. [b] (\lambda v_b. v_a \ v_b \ k))
\]

We bind \([a]\) to a variable \(v_a\), then bind \([b]\) to a variable \(v_b\), then perform the application \(v_a \ v_b\).
Interface of a monad

A monad is defined by a parameterized type \( \alpha \text{ mon} \) and operations \text{ret}, \text{bind} and \text{run}, with types:

\[
\begin{align*}
\text{ret} & : \forall \alpha. \alpha \rightarrow \alpha \text{ mon} \\
\text{bind} & : \forall \alpha, \beta. \alpha \text{ mon} \rightarrow (\alpha \rightarrow \beta \text{ mon}) \rightarrow \beta \text{ mon} \\
\text{run} & : \forall \alpha. \alpha \text{ mon} \rightarrow \alpha
\end{align*}
\]

The type \( \tau \text{ mon} \) is the type of computations that eventually produce a value of type \( \tau \).

\text{ret} \ a \text{ encapsulates a pure expression } a : \tau \text{ as a trivial computation (of type } \tau \text{ mon}) \text{ that immediately produces the value of } a.

\text{bind} \ a \ (\lambda x. b) \text{ performs the computation } a : \tau \text{ mon}, \text{ binds its value to } x : \tau, \text{ then performs the computation } b : \tau' \text{ mon}.

\text{run} \ a \text{ is the execution of a monadic program } a, \text{ extracting its return value.}
Monadic laws

The \texttt{ret} and \texttt{bind} operations of the monad are supposed to satisfy the following algebraic laws:

\[
\begin{align*}
\text{bind } (\text{ret } a) f & \approx f \ a \\
\text{bind } a (\lambda x. \text{ret } x) & \approx a \\
\text{bind } (\text{bind } a (\lambda x. b)) (\lambda y. c) & \approx \text{bind } a (\lambda x. \text{bind } b (\lambda y. c))
\end{align*}
\]

The relation \( \approx \) needs to be made more precise, but intuitively means “behaves identically”.
Example: the Exception monad
(also called the Error monad)

```
type α mon = V of α | E of exn

ret a = V(a)

bind m f = match m with E(x) -> E(x) | V(x) -> f x

run m = match m with V(x) -> x
```

*bind* encapsulates the propagation of exceptions in compound expressions such as *a b* or *let* bindings.

Additional operations in this monad:

```
raise x = E(x)

trywith m f = match m with E(x) -> f x | V(x) -> V(x)
```
Example: the State monad

\[
\text{type } \alpha \text{ mon } = \text{state } \rightarrow \alpha \times \text{state}
\]

\[
\text{ret a } = \lambda s. (a, s)
\]

\[
\text{bind } m \ f = \lambda s. \text{match } m \ s \ \text{with } (x, s') \rightarrow f \ x \ s'
\]

\[
\text{run } m = \text{match } m \ \text{empty\_store} \ \text{with } (x, s) \rightarrow x
\]

bind encapsulates the threading of the state in compound expressions.

Additional operations in this monad:

\[
\text{ref } x = \lambda s. \text{store\_alloc } x \ s
\]

\[
\text{deref } r = \lambda s. (\text{store\_read } r \ s, s)
\]

\[
\text{assign } r \ x = \lambda s. ((), \text{store\_write } r \ x \ s)
\]
Example: the Continuation monad

\[
\text{type } \alpha \text{ mon } = (\alpha \to \text{answer}) \to \text{answer}
\]

\[
\text{ret } a = \lambda k. k \ a
\]

\[
\text{bind } m \ f = \lambda k. m (\lambda v. f \ v \ k)
\]

\[
\text{run } m = m (\lambda x. x)
\]

Additional operations in this monad:

\[
\text{callcc } f = \lambda k. f \ k \ k
\]

\[
\text{throw } x \ y = \lambda k. x \ y
\]
Alternate presentation of a monad

The alternate presentation replaces bind by two operations fmap and join:

\[
\begin{align*}
\text{ret} & : \forall \alpha. \alpha \to \alpha \text{ mon} \\
\text{fmap} & : \forall \alpha, \beta. (\alpha \to \beta) \to (\alpha \text{ mon} \to \beta \text{ mon}) \\
\text{join} & : \forall \alpha, (\alpha \text{ mon}) \text{ mon} \to \alpha \text{ mon}
\end{align*}
\]

The two presentations are related as follows:

\[
\begin{align*}
\text{bind } a f & \equiv \text{join}(\text{fmap } f \ a) \\
\text{fmap } f \ m & \equiv \text{bind } m (\lambda x. \text{ret}(f \ x)) \\
\text{join } mm & \equiv \text{bind } mm (\lambda m. m)
\end{align*}
\]

The alternate presentation is closer to category theory but less convenient for programming,
Outline

1. Introduction to monads

2. The monadic translation
   - Definition
   - Correctness
   - Application to some monads

3. Monadic programming
   - More examples of monads
   - Monad transformers

4. Bonus track: applicative structures

5. Bonus track: comonads
The monadic translation

Core constructs

\[
\begin{align*}
[N] &= \text{ret } N \\
[x] &= \text{ret } x \\
[\lambda x. a] &= \text{ret } (\lambda x.[a]) \\
[\text{let } x = a \text{ in } b] &= \text{bind } [a] (\lambda x.[b]) \\
[a \ b] &= \text{bind } [a] (\lambda v_a. \text{bind } [b] (\lambda v_b. v_a \ v_b))
\end{align*}
\]

These translation rules are shared between all monads.

Effect on types: if \( a : \tau \) then \([a] : [\tau] \text{ mon}\)

where \([\tau_1 \rightarrow \tau_2] = [\tau_1] \rightarrow [\tau_2] \text{ mon}\) and \([\tau] = \tau\) for base types \( \tau \).
The monadic translation

Extensions

\[
\begin{align*}
\sem{\mu f. \lambda x. a} & = \ \text{ret} \ (\mu f. \lambda x. \sem{a}) \\
\sem{a \ op \ b} & = \ \text{bind} \ \sem{a} \ (\lambda v_a. \ \text{bind} \ \sem{b} \ (\lambda v_b. \ \text{ret} \ (v_a \ op \ v_b))) \\
\sem{C(a_1, \ldots, a_n)} & = \ \text{bind} \ \sem{a_1} \ (\lambda v_1. \ \ldots \ \text{bind} \ \sem{a_n} \ (\lambda v_n. \ \text{ret}(C(v_1, \ldots, v_n))))
\end{align*}
\]
Example of monadic translation

\[ [1 + f \ x] \]
\[
= \text{bind (ret 1) (λv1.}
\]
\[
\text{bind (bind (ret f) (λv2.}
\]
\[
\text{bind (ret x) (λv3. v2 v3))) (λv4.}
\]
\[
\text{ret (v1 + v4))}
\]

After administrative reductions using the first monadic law:

\[ [1 + f \ x] \]
\[
= \text{bind (f x) (λv. ret (1 + v))}
\]
Example of monadic translation

\[
[\mu \text{fact. } \lambda n. \text{ if } n = 0 \text{ then } 1 \text{ else } n \ast \text{fact}(n-1)] = \\
\text{ret } (\mu \text{fact. } \lambda n. \\
\text{ if } n = 0 \\
\text{ then ret 1 } \\
\text{ else bind (fact}(n-1)) (\lambda v. \text{ret } (n \ast v))
\]
The monadic translation

Monad-specific constructs and operations

Most additional constructs for exceptions, state and continuations can be treated as regular function applications of the corresponding additional operations of the monad. For instance, in the case of \texttt{raise a}:

\[
[\text{raise } a] = \text{bind } (\text{ret raise}) (\lambda v_r. \text{bind } [a] (\lambda v_a. v_r v_a))
\]

The bind takes care of propagating exceptions raised in \texttt{a}.

The only case where we need a special translation rule is the the \texttt{try...with} construct:

\[
[\text{try } a \text{ with } x \rightarrow b] = \text{trywith } [a] (\lambda x. [b])
\]
The monadic translation

Correctness

Syntactic properties of the monadic translation

Define the monadic translation of a value \([v]_v\) as follows:

\[
[N]_v = N \quad [\lambda x. a]_v = \lambda x.[a]
\]

Lemma 1 (Translation of values)

\([v] = \text{ret} \ [v]_v\) for all values \(v\). Moreover, \([v]_v\) is a value.

Lemma 2 (Monadic substitution)

\([a[x \leftarrow v]] = [a][x \leftarrow [v]_v]\) for all values \(v\).
Reasoning about reductions of the translations

If a reduces, is it the case that the translation $[a]$ reduces? This depends on the monad:

- For the exception monad, this is true.
- For the state and continuation monads, $[a]$ is a $\lambda$-abstraction which cannot reduce.

To reason about the evaluation of $[a]$, we need in general to put this term in an appropriate context, for instance

- For the state monad: $[a] s$ where $s$ is a store value.
- For the continuation monad: $[a] k$ where $k$ is a continuation $\lambda x \ldots$
Contextual equivalence

To overcome this problem, we assume that the monad defines an equivalence relation $a \approx a'$ between terms, which is reflexive, symmetric and transitive, and satisfies the following properties:

1. $(\lambda x.a) \approx a[x \leftarrow v]$  \hspace{2cm} ($\beta_v$ reduction)
2. $\text{bind} \ (\text{ret} \ v) \ (\lambda x.b) \approx b[x \leftarrow v]$ \hspace{2cm} (first monadic law)
3. $\text{bind} \ a \ (\lambda x.b) \approx \text{bind} \ a' \ (\lambda x.b)$ if $a \approx a'$ \hspace{2cm} (compat. context)
4. If $a \approx \text{ret} \ v$, then $\text{run} \ a \rightarrow^* v$. 

X. Leroy (INRIA)
Correctness of the monadic translation

**Theorem 3**

If $a \Rightarrow v$, then $\llbracket a \rrbracket \approx \text{ret } \llbracket v \rrbracket_v$.

The proof is by induction on a derivation of $a \Rightarrow v$ and case analysis on the last evaluation rule.

The cases $a = N$, $a = x$ and $a = \lambda x. b$ are obvious: we have $a = v$, therefore $\llbracket a \rrbracket = \text{ret } \llbracket v \rrbracket_v$. 
Correctness of the monadic translation

For the let case:

\[
\begin{align*}
    b \implies v' &\quad c[x \leftarrow v'] \implies v \\
\hline
    \text{let } x = b \text{ in } c \implies v
\end{align*}
\]

The following equivalences hold:

\[
\begin{align*}
    \llbracket a \rrbracket & = \text{bind } \llbracket b \rrbracket (\lambda x.\llbracket c \rrbracket) \\
    (\text{ind.hyp + prop.3}) & \approx \text{bind } (\text{ret } \llbracket v' \rrbracket_v) (\lambda x.\llbracket c \rrbracket) \\
    (\text{prop.2}) & \approx \llbracket c \rrbracket[x \leftarrow \llbracket v' \rrbracket_v] = \llbracket c[x \leftarrow v'] \rrbracket \\
    (\text{ind.hyp.}) & \approx \text{ret } \llbracket v \rrbracket_v
\end{align*}
\]
Correctness of the monadic translation

For the application case:

\[ b \Rightarrow \lambda x. d \quad c \Rightarrow v' \quad d[x \leftarrow v'] \Rightarrow v \]

\[ b\ c \Rightarrow v \]

The following equivalences hold:

\[ \llbracket a \rrbracket = \text{bind } \llbracket b \rrbracket (\lambda y. \text{bind } \llbracket c \rrbracket (\lambda z. y z)) \]

(ind.hyp + prop.3) \quad \approx \quad \text{bind } (\text{ret } (\lambda x. \llbracket d \rrbracket)) (\lambda y. \text{bind } \llbracket c \rrbracket (\lambda z. y z)) \]

(prop.2) \quad \approx \quad \text{bind } \llbracket c \rrbracket (\lambda z. (\lambda x. \llbracket d \rrbracket) z)) \]

(ind.hyp + prop.3) \quad \approx \quad \text{bind } (\text{ret } \llbracket v' \rrbracket_v (\lambda z. (\lambda x. \llbracket d \rrbracket) z)) \]

(prop.2) \quad \approx \quad (\lambda x. \llbracket d \rrbracket) \llbracket v' \rrbracket_v \]

(prop.1) \quad \approx \quad \llbracket d \rrbracket [x \leftarrow \llbracket v' \rrbracket_v] = \llbracket d [x \leftarrow v] \rrbracket \]

(ind.hyp.) \quad \approx \quad \text{ret } \llbracket v \rrbracket_v \]
Correctness of the monadic translation

Theorem 4

If $a \Rightarrow v$, then $\text{run } [a] \rightarrow^* [v]_v$.

Proof.

Follows from theorem 3 and property 4 of $\approx$.

Note that we proved this theorem only for pure terms $a$ that do not use monad-specific constructs. These constructs add more cases, but often the proof cases for application, etc, are unchanged. (Exercise.)
Application to the Exception monad

Define $a_1 \approx a_2$ as $\exists a, \ a_1 \rightarrow a \leftarrow a_2$.

Some interesting properties of this relation:

- If $a \rightarrow a'$ then $a \approx a'$.
- If $a \approx a'$ and $a \rightarrow^* v$, then $a' \rightarrow^* v$.
- It is transitive, for if $a_1 \rightarrow^* a \leftarrow a_2 \rightarrow^* a' \leftarrow a_3$, determinism of the $\rightarrow$ reduction implies that either $a \rightarrow^* a'$ or $a' \rightarrow^* a$. In the former case, $a_1 \rightarrow^* a' \leftarrow a_3$, and in the latter case, $a_1 \rightarrow^* a \leftarrow a_3$.
- It is compatible with reduction contexts: $E[a_1] \approx E[a_2]$ if $a_1 \approx a_2$ and $E$ is a reduction context.

We now check that $\approx$ satisfies the hypothesis of theorem 3.
Application to the Exception monad

1. \((\lambda \mathbf{a}. \mathbf{b}) \mathbf{v} \approx \mathbf{a}[\mathbf{x} \leftarrow \mathbf{v}]\)
   Trivial since \((\lambda \mathbf{a}. \mathbf{b}) \mathbf{v} \to \mathbf{a}[\mathbf{x} \leftarrow \mathbf{v}]\).

2. \(\text{bind} \ (\text{ret} \ \mathbf{v}) \ (\lambda \mathbf{x}. \mathbf{b}) \approx \mathbf{b}[\mathbf{x} \leftarrow \mathbf{v}]\). We have
   \[
   \begin{align*}
   \text{bind} \ (\text{ret} \ \mathbf{v}) \ (\lambda \mathbf{x}. \mathbf{b}) \\
   \to & \quad \text{bind} \ (\mathbf{V}(\mathbf{v})) \ (\lambda \mathbf{x}. \mathbf{b}) \\
   \to & \quad \text{match} \ \mathbf{V}(\mathbf{v}) \ 	ext{with} \ E(\mathbf{y}) \to \mathbf{y} \ | \ \mathbf{V}(\mathbf{z}) \to (\lambda \mathbf{x}. \mathbf{b}) \ \mathbf{z} \\
   \to & \quad (\lambda \mathbf{x}. \mathbf{b}) \ \mathbf{v} \to \mathbf{b}[\mathbf{x} \leftarrow \mathbf{v}]
   \end{align*}
   \]

3. \(\text{bind} \ \mathbf{a}_1 \ (\lambda \mathbf{x}. \mathbf{b}) \approx \text{bind} \ \mathbf{a}_2 \ (\lambda \mathbf{x}. \mathbf{b}) \) if \(\mathbf{a}_1 \approx \mathbf{a}_2\).
   Trivial since \(\text{bind} \ [\] \ (\lambda \mathbf{x}. \mathbf{b})\) is an evaluation context.

4. If \(\mathbf{a} \approx \text{ret} \ \mathbf{v}\), then \(\text{run} \ \mathbf{a} \to^* \mathbf{v}\).
   Since \(\text{ret} \ \mathbf{v} \to^* \mathbf{V}(\mathbf{v})\), we have \(\mathbf{a} \to^* \mathbf{V}(\mathbf{v})\) and the result follows.
The monadic translation

Application to some monads

Application to the Continuation monad

Define $a_1 \approx a_2$ as $\forall k \in \text{Values}, \ \exists a, \ a_1 \ k \rightarrow a \leftarrow a_2 \ k$.

1. $(\lambda x.a) \ v \approx a[x \leftarrow v]$  
   Trivial since $(\lambda x.a) \ v \ k \rightarrow a[x \leftarrow v] \ k$.

2. $\text{bind} \ (\text{ret} \ v) \ (\lambda x.b) \approx b[x \leftarrow v]$. We have

   \[
   \text{bind} \ (\text{ret} \ v) \ (\lambda x.b) \ k \rightarrow \text{bind} \ (\lambda k'. k' \ v) \ (\lambda x.b) \\
   \rightarrow^* (\lambda k'. k' \ v) \ (\lambda y. (\lambda x.b) \ y \ k) \\
   \rightarrow (\lambda y. (\lambda x.b) \ y \ k) \ v \\
   \rightarrow (\lambda x.b) \ v \ k \\
   \rightarrow b[x \leftarrow v] \ k
   \]
Application to the Continuation monad

1. 
   \[ \text{bind } a_1 (\lambda x. b) \approx \text{bind } a_2 (\lambda x. b) \text{ if } a_1 \approx a_2 \]
   
   We have \( \text{bind } a_i (\lambda x. b) k \xrightarrow{\ast} a_i (\lambda v. (\lambda x. b) v k) \) for \( i = 1, 2 \).
   
   Using the hypothesis \( a_1 \approx a_2 \) with the continuation \( (\lambda v. (\lambda x. b) v k) \),
   
   we obtain a term \( a \) such that \( a_i (\lambda v. (\lambda x. b) v k) \xrightarrow{\ast} a \) for \( i = 1, 2 \).
   
   Therefore, \( \text{bind } a_i (\lambda x. b) k \xrightarrow{\ast} a \) for \( i = 1, 2 \), and the result follows.

2. 
   If \( a \approx \text{ret } v \), then \( \text{run } a \xrightarrow{\ast} v \).
   
   The result follows from \( \text{ret } v (\lambda x. x) \xrightarrow{\ast} v \).
Application to the State monad

Define $a_1 \approx a_2$ as $\forall s \in Values, \exists a, a_1 s \rightarrow a \leftarrow a_2 s$.

The proofs of hypotheses 1–4 are similar to those for exceptions.
Outline

1 Introduction to monads

2 The monadic translation
   - Definition
   - Correctness
   - Application to some monads

3 Monadic programming
   - More examples of monads
   - Monad transformers

4 Bonus track: applicative structures

5 Bonus track: comonads
Monads as a general programming technique

Monads provide a systematic way to structure programs into two well-separated parts:

- the algorithms proper, and
- the “plumbing” of computations needed by these algorithms (state passing, exception handling, non-deterministic choice, etc). The “plumbing” can often be hidden inside a reusable library.

In addition, monads can also be used to modularize code and offer new possibilities for reuse:

- Code in monadic form can be parameterized over a monad and reused with several monads.
- Monads themselves can be built in an incremental manner.
The Counting monad (a.k.a. the Complexity monad)

Counts how many times the \texttt{tick} monadic operation is evaluated during execution. A special case of the State monad, with only one integer reference that can only be incremented.

\texttt{module Count = struct}
\begin{verbatim}
  type \alpha \ mon = int \rightarrow \alpha \times int

  let ret a = fun n -> (a, n)
  let bind m f = fun n -> match m n with (x, n') -> f x n'
  let run m = m 0

  let tick m = fun n -> m (n+1)
end}
\end{verbatim}

\textbf{Infix notation:} $m >>= f$ for \texttt{bind m f}.
Example of use

Before monadic translation: (counts the number of comparisons)

```ocaml
let rec insert x l =
    match l with
    | [] -> [x]
    | h :: t -> if tick(x < h) then x :: l else h :: insert x t
```

After monadic translation:

```ocaml
let rec insert x l =
    match l with
    | [] -> Count.ret [x]
    | h :: t ->
        Count.(tick (ret (x < h)) >>= fun b ->
            if b
            then ret (x::l)
            else insert x t >>= fun r -> ret (h::r))
```
The Logging monad (a.k.a. the Writer monad)

Enables computations to log messages. A generalization of the Counting monad, where a list of messages is maintained instead of a counter.

module Log = struct
  type log = string list
  type α mon = log → α × log

  let ret a = fun l -> (a, l)
  let bind m f = fun l -> match m l with (x, l') -> f x l'
  let run m = match m [] with (x, l) -> (x, List.rev l)

  let log msg = fun l -> ((), msg :: l)
end
Example of use

Before monadic translation:

```ocaml
let abs n =
  if n >= 0
  then (log "positive"; n)
  else (log "negative"; -n)
```

After monadic translation:

```ocaml
let abs n =
  if n >= 0
  then Log.(log "positive" >>= fun _ -> ret n)
  else Log.(log "negative" >>= fun _ -> ret (-n))
```
The Environment monad (a.k.a. the Reader monad)

Propagate an environment “down” all branches of a computation.

```ocaml
module Environment = struct
  type α mon = env -> α

  let ret x = fun e -> x
  let bind m f = fun e -> f (m e) e
  let run m = m initial_env

  let getenv varname =
    fun e -> map_lookup varname e
end
```
Non-determinism, a.k.a. the List monad

Provides computations with non-deterministic choice as well as failure. Underneath, computes the list of all possible results.

```ocaml
module Nondet = struct
    type α mon = α list

    let ret a = a :: []
    let rec bind m f =
        match m with [] -> [] | hd :: tl -> f hd @ bind tl f
    let run m = match m with hd :: tl -> hd
    let runall m = m
    let fail = []
    let either a b = a @ b
end
```
Example of use

All possible ways to insert an element \( x \) in a list \( l \):

\[
\text{let rec insert } x \ l = \\
\text{Nondet.}(\text{either} \\
\text{(ret } (x :: l)) \\
\text{(match } l \text{ with} \\
\mid [] \rightarrow \text{fail} \\
\mid \text{hd :: tl} \rightarrow \text{insert } x \ tl \ >>\ >>= \ fun \ l' \rightarrow \text{ret } (\text{hd :: l'}))\)
\]

All permutations of a list \( l \):

\[
\text{let rec permut } l = \\
\text{match } l \text{ with} \\
\mid [] \rightarrow \text{Nondet.} \text{ret } [] \\
\mid \text{hd :: tl} \rightarrow \text{Nondet.} (\text{permut } tl \ >>\ >>= \ fun \ l' \rightarrow \text{insert } \text{hd } l')
\]
The Parsing monad

A variant of the state monad where the state is the input text that remains to be parsed. Supports failure like the Exception monad.

module Parsing = struct

  type α result =
  | Success of α * char list
  | Failure

  type α mon = char list -> α result

  let ret (x: α): α mon = fun txt -> Success(x, txt)

  let bind (m: α mon) (f: α -> β mon): β mon =
    fun txt ->
      match m txt with
      | Failure -> Failure
      | Success(x, txt’) -> f x txt’
The Parsing monad

Specific operations in this monad: symbol \( c \) (recognize and consume the single character \( c \)) and either \( m_1 \) \( m_2 \) (alternative with backtracking).

```ocaml
define symbol c : char mon =
  fun txt ->
    match txt with
    | [] -> Failure
    | c' :: txt' -> if c' = c then Success(c, txt') else Failure

define either (m1: \( \alpha \) mon) (m2: \( \alpha \) mon): \( \alpha \) mon =
  fun txt ->
    match m1 txt with
    | Failure -> m2 txt
    | Success(x, txt') as res -> res
```

X. Leroy (INRIA)
The Parsing monad

Some derived operations in this monad: 0 or 1 (opt), 0 or 1 or several (star), and 1 or several (plus) occurrences of a given recognizer $m$.

```ocaml
let opt (m: α mon): α option mon =  
either (m >>= fun x -> ret (Some x)) (ret None)

let rec star (m: α mon): α list mon =  
either (plus m) (ret [])

and plus (m: α mon): α list mon =  
m >>= fun x ->  
star m >>= fun y ->  
ret (x :: y)
```
Monads for randomized computations

Consider a source language with randomized constructs such as

\[
\text{rand } n \quad \text{return a uniformly-distributed integer in } [0, n[ \\
\text{choose } p \, a \, b \quad \text{evaluate either } a \text{ with probability } p \in [0, 1] \text{ or } b \text{ with probability } 1 - p
\]

In a monadic interpretation, these constructs have type

\[
\text{rand} : \text{int} \rightarrow \text{int mon} \\
\text{choose} : \forall \alpha. \text{float} \rightarrow \alpha \text{ mon} \rightarrow \alpha \text{ mon} \rightarrow \alpha \text{ mon}
\]
Examples of randomized computations

let dice num_sides =
    M.(rand num_sides >>= fun n -> ret (n + 1))

let roll_3d6 =
    M.(dice 6 >>= fun d1 ->
        dice 6 >>= fun d2 ->
        dice 6 >>= fun d3 ->
        ret (d1 + d2 + d3))

let traffic_light =
    M.choose 0.05 (M.ret Yellow)
    (M.choose 0.5 (M.ret Red)
        (M.ret Green))
First implementation: the Simulation monad

Uses a pseudo-random number generator to give values to random variables (Monte-Carlo simulation). This is a variant of the State monad.

```ocaml
module Random_Simulation = struct
  type α mon = int → α × int

  let ret a = fun s -> (a, s)
  let bind m f = fun s -> match m s with (x, s) -> f x s

  let next_state s = s * 25173 + 1725
  let rand n = fun s -> ((abs s) mod n, next_state s)
  let choose p a b = fun s ->
    if float (abs s) <= p *. float max_int
    then a (next_state s) else b (next_state s)
end
```
Second implementation: the Distribution monad

With the same interface, this monad computes the distribution of the results: all possible result values along with their probabilities. This is an extension of the List monad.

```ocaml
module Random_Distribution = struct
  type α mon = (α × float) list

  let ret a = [(a, 1.0)]
  let bind m f = [ (y, p1 *. p2) | (x, p1) <- m, (y, p2) <- f x ]

  let rand n = [ (0, \frac{1}{n}); \ldots; (n-1, \frac{1}{n}) ]
  let choose p a b = [ (x, p *. p1) | (x, p1) <- a ] @
                         [ (x, (1.0 -. p) *. p2) | (x, p2) <- b ]
end
```
Third implementation: the Expectation monad

Still with the same interface, this monad computes the expectation of a result (of type $\alpha$) w.r.t. a given measure (a function $\alpha \to \text{float}$). This is an extension of the Continuation monad.

module Random.Expectation = struct
  type $\alpha$ mon = ($\alpha$ -> float) -> float

  let ret x = fun k -> k x
  let bind x f = fun k -> x (fun vx -> f vx k)

  let rand n = fun k -> \( \frac{1}{n} \) * k 0 + ... + \( \frac{1}{n} \) * k (n-1)

  let choose p a b = fun k -> p * a k + (1.0 -. p) * b k
end
Combining monads

What if we need both exceptions and state in an algorithm? We can write (from scratch) a monad that supports both. Notice that there are several choices:

- type \( \alpha \text{ mon} = \text{state} \to (\alpha \times \text{state}) \text{ outcome} \)
  l.e. the state is discarded when we raise an exception.
- type \( \alpha \text{ mon} = \text{state} \to \alpha \text{ outcome} \times \text{state} \)
  l.e. the state is kept when we raise an exception.

In the second case, \texttt{trywith} can be defined in two ways:

\[
\text{trywith } m \ f \ = \ \lambda s. \ \text{match } m \ s \ \text{with} \\
\mid (V(v), s') \to (V(v), s') \\
\mid (E(e), s') \to f \ e \left( \begin{array}{c} s \\ s' \end{array} \right)
\]

The \( s \) choice backtracks the assignments made by the computation \( m \); the \( s' \) choice preserves them.
Composing two monads?

Given two monads

\[
\begin{align*}
\text{type } \alpha \text{ mon1} & \quad \text{type } \alpha \text{ mon2} \\
\text{ret1: } \alpha & \rightarrow \alpha \text{ mon1} & \text{ret2: } \alpha & \rightarrow \alpha \text{ mon2} \\
\text{bind1: } \alpha \text{ mon1} & \rightarrow (\alpha \rightarrow \beta \text{ mon1}) & \text{bind2: } \alpha \text{ mon2} & \rightarrow (\alpha \rightarrow \beta \text{ mon2}) \\
& \rightarrow \beta \text{ mon1} & & \rightarrow \beta \text{ mon2}
\end{align*}
\]

is there a generic way to \textit{compose} them? Let’s try...

\[
\begin{align*}
\text{type } \alpha \text{ mon} &= \alpha \text{ mon1} \text{ mon2} \\
\text{let ret } (x : \alpha) : \alpha \text{ mon} &= \text{ret2} (\text{ret1 } x) \\
\text{let bind } (x : \alpha \text{ mon}) (f : \alpha \rightarrow \beta \text{ mon}) : \beta \text{ mon} &= \\
& \quad \text{bind2 } x \ (\lambda y : \alpha \text{ mon1} \cdot \text{ret2} ( \\
& \quad \quad \text{bind1 } y \ (\lambda z : \alpha \cdot ???? (f \ z))))
\end{align*}
\]
Composing two monads?

Without additional operations provided by the second (outer) monad, there is no way to define the bind of the composed monad:

\[
\text{let bind } (x : \alpha \text{ mon}) (f : \alpha \to \beta \text{ mon}) : \beta \text{ mon} = \\
\text{bind2 } x (\lambda y : \alpha \text{ mon1. ret2 (}} \\
\quad \text{bind1 } y (\lambda z : \alpha. ???? (f z)))
\]

Since \(f z : \beta \text{ mon1 mon2}\) and \(\text{bind1}\) demands something of type \(\beta \text{ mon1}\), we need a term \(????\) of type \(\beta \text{ mon1 mon2} \to \beta \text{ mon1}\).

It is impossible to construct a closed, terminating term of this type just from the \(\text{ret2}\) and \(\text{bind2}\) operations of the second monad!
Monad transformers

A monad transformer takes any monad $M$ and returns a monad $M'$ with additional capabilities, e.g. exceptions, state, continuation. It also provides a lift function that transforms $M$ computations (of type $\alpha \ M.\ mon$) into $M'$ computations (of type $\alpha \ M'.\ mon$).

In Caml, monad transformers are naturally presented as functors, i.e. functions from modules to modules. (Haskell uses type classes.)
Signature for monads

The Caml module signature for a monad is:

module type MONAD = sig
  type α mon
  val ret: α -> α mon
  val bind: α mon -> (α -> β mon) -> β mon
  val run: α mon -> α
end
The Identity monad

The Identity monad is a trivial instance of this signature:

```ocaml
module Identity = struct
  type α mon = α
  let ret x = x
  let bind m f = f m
  let run m = m
end
```
Monad transformer for exceptions

module ExceptionTransf(M: MONAD) = struct
    type α outcome = V of α | E of exn
    type α mon = (α outcome) M.mon

    let ret x = M.ret (V x)
    let bind m f =
        M.bind m (function E e -> M.ret (E e) | V v -> f v)
    let lift x = M.bind x (fun v -> M.ret (V v))
    let run m = M.run (M.bind m (function V x -> M.ret x))

    let raise e = M.ret (E e)
    let trywith m f =
        M.bind m (function E e -> f e | V v -> M.ret (V v))
end
Monad transformer for state

module StateTransf(M: MONAD) = struct
  type α mon = state -> (α * state) M.mon

  let ret x = fun s -> M.ret (x, s)
  let bind m f = 
    fun s -> M.bind (m s) (fun (x, s') -> f x s')
  let lift m = fun s -> M.bind m (fun x -> M.ret (x, s))
  let run m = 
    M.run (M.bind (m empty_store) (fun (x, s') -> M.ret x))

  let ref x = fun s -> M.ret (store_alloc x s)
  let deref r = fun s -> M.ret (store_read r s, s)
  let assign r x = fun s -> M.ret (store_write r x s)
end
Monad transformer for continuations

module ContTransf(M: MONAD) = struct
  type α mon = (α -> answer M.mon) -> answer M.mon

  let ret x = fun k -> k x
  let bind m f = fun k -> m (fun v -> f v k)
  let lift m = fun k -> M.bind m k
  let run m = M.run (m (fun x -> M.ret x))

  let callcc f = fun k -> f k k
  let throw c x = fun k -> c x
end
Using monad transformers

module StateAndException = struct
    include ExceptionTransf(State)
    let ref x = lift (State.ref x)
    let deref r = lift (State.deref r)
    let assign r x = lift (State.assign r x)
end

This gives a type $\alpha \text{ mon } = \text{state} \rightarrow \alpha \text{ outcome } \times \text{state}$, i.e. state is preserved when raising exceptions.

The other combination, StateTransf(Exception) gives $\alpha \text{ mon } = \text{state} \rightarrow (\alpha \times \text{state}) \text{ outcome}$, i.e. state is discarded when an exception is raised.
The Concurrency monad transformer

(Based on an approach by Tomas Petricek, 2011)

Given any monad $M$, we define concurrency (interleaving of computations) via the following type of resumptions:

```ocaml
module Concur(M: MONAD) = struct

  type α mon =
  | Done of α
  | Step of (α mon) M.mon

A resumption describes a sequence of computations in monad $M$:

- $\text{Done } v$ denotes no computations and a final result value $v$
- $\text{Step } m$ denotes the computation $m$ followed by the resumption that $m$ returns as its value.
```
The Concurrency monad transformer

type \( \alpha \) mon = Done of \( \alpha \) | Step of (\( \alpha \) mon) M.mon

A resumption can trivially be turned into a single computation in monad \( M \), then run:

```ml
let rec perform (x: \( \alpha \) mon): \( \alpha \) M.mon =
    match x with
    | Done res -> M.ret res
    | Step m -> M.bind m perform

let run (x: \( \alpha \) mon) = M.run (perform x)
```

However, by keeping the list-like structure of resumptions, we are able to interleave two resumptions, simulating concurrent execution.
The Concurrency monad transformer

The `ret` operation of the Concurrency monad performs zero computations:

```plaintext
let ret (x: α): α mon = Done x
```

The `act` operation (also known as `lift`) performs just one computation:

```plaintext
let act (m: α M.mon): α mon = 
    Step (M.bind m (fun res -> M.ret (Done res)))
```

The `bind` operation is similar to list concatenation, appending two lists of computations:

```plaintext
let rec bind (m: α mon) (f: α -> β mon): β mon = 
    match m with 
    | Done res -> f res 
    | Step s -> Step (M.bind s (fun m’ -> M.ret (bind m’ f)))
```
The Concurrency monad transformer

Finally, \texttt{par} interleaves the computations of two resumptions:

\begin{verbatim}
let rec par (m1: α mon) (m2: β mon) : (α * β) mon =
    match m1, m2 with
    | Done r1, Done r2 -> Done (r1, r2)
    | Step s1, Step s2 ->
        Step (M.bind s1 (fun m1' ->
            M.bind s2 (fun m2' -> M.ret (par m1' m2'))))
    | Done r1, Step s2 ->
        Step (M.bind s2 (fun m2' -> M.ret (par (Done r1) m2')))
    | Step s1, Done r2 ->
        Step (M.bind s1 (fun m1' -> M.ret (par m1' (Done r2))))
\end{verbatim}
Example of use

module M = Concur(Log)

let rec loop n s =
  if n <= 0
  then M.ret ()
  else M.(act (Log.log s) >>= fun _ -> loop (n-1) s)

M.(run (act (Log.log "start:")) >>= fun _ ->
  par (loop 6 "a") (loop 4 "b")))

This code will log “start:ababababaa”
Outline

1. Introduction to monads
2. The monadic translation
   - Definition
   - Correctness
   - Application to some monads
3. Monadic programming
   - More examples of monads
   - Monad transformers
4. Bonus track: applicative structures
5. Bonus track: comonads
Applicative structures

Monads impose a very “sequential” programming style, where all subcomputations are explicitly sequenced. This is heavy style if the monad has no effects or effects that commute (e.g. read effects).

Example: evaluating expressions containing variables using the Environment monad.

**In direct style:**

```latex
let rec eval = function
    | Const n -> n
    | Var v -> getenv v
    | Plus(a, b) ->
        eval a + eval b
```

**In monadic style:**

```latex
let rec eval = function
    | Const n -> ret n
    | Var v -> getenv v
    | Plus(a, b) ->
        eval a >>= fun n ->
        eval b >>= fun m ->
        ret (n + m)
```
Monadic application combinators

This can be improved by defining application combinators:

```ml
let mapp (f: α -> β) (m: α mon ) : β mon =
    bind m (fun x -> ret (f x))
```

```ml
let mapp2 (f: α -> β -> γ) (m1: α mon ) (m2: β mon ): γ mon =
    bind m1 (fun x1 -> bind m2 (fun x2 -> ret (f x1 x2)))
```

(etc). The eval function becomes nicer:

```ml
let rec eval = function
    ...
    | Plus(a, b) -> mapp2 (+) (eval a) (eval b)
```
The monadic application combinator

We can avoid defining a combinator for every arity, as follows:

\[
\text{let } \langle \ast \rangle \ (f : (\alpha \to \beta) \text{ mon}) \ (m : \alpha \text{ mon}) : \beta \text{ mon } = \\
\text{bind } f \ (\text{fun } vf \to \text{bind } m \ (\text{fun } vm \to \text{ret } (vf \ vm)))
\]

Using the fact that \(\langle \ast \rangle\) associates to the left in OCaml, we get:

\[
\text{let rec eval } = \text{ function} \\
\quad \ldots \\
\quad | \text{Plus}(a, b) \to \text{ret } (+) \langle \ast \rangle \text{ eval } a \langle \ast \rangle \text{ eval } b
\]

More generally: \(\text{ret } f \langle \ast \rangle m_1 \langle \ast \rangle \ldots \langle \ast \rangle m_n\) denotes the application of the pure function \(f\) to the results of the monadic computations \(m_1, \ldots, m_n\).
Applicative structures


An applicative structure is a parameterized type $\alpha \text{ app}$ of effectful computations producing a value of type $\alpha$, plus two operations:

\[
\text{pure} : \forall \alpha, \alpha \to \alpha \text{ app} \\
\text{<*>} : \forall \alpha \beta, (\alpha \to \beta) \text{ app} \to \alpha \text{ app} \to \beta \text{ app}
\]

pure embeds values (computations without effects) into computations.

<*> , pronounced “apply”, performs function application with propagation of effects.
The laws of applicative structures

\[
pure (\lambda x. x) \triangleright u \approx u
\]
\[
pure (\lambda f. \lambda g. \lambda x. f(g x)) \triangleright u \triangleright v \triangleright w \approx u \triangleright (v \triangleright w)
\]
\[
pure f \triangleright pure x \approx pure(f x)
\]
\[
u \triangleright pure x \approx pure(\lambda f. f x) \triangleright u
\]

Intuitively: we can reorder/simplify pure computations, as long as the order of effectful computations is preserved.

Categorically: a strong lax monoidal functor...
Monads and applicative structures

Every monad $M$ defines an applicative structure:

```haskell
type $\alpha$ app = $\alpha$ M.mon

let pure x = M.ret x

let $\langle*\rangle$ f x =
    M.bind f (fun vf -> M.bind x (fun vx -> M.ret (vf vx)))
```

(This is for left-to-right application. Can also do right-to-left by swapping the two `bind`.

However:

- Sometimes, other definitions of $\langle*\rangle$ are more useful.
- Some types are not monads but have a useful applicative structure.
Monads with nonstandard application

Consider the Exception monad. We’d like to collect all the exceptions raised during the evaluation of an expression, not just the first one:

\[(\text{raise } A) + (\text{raise } B) \quad \longrightarrow \quad \text{Uncaught exceptions: } A, B\]

Let us redefine the type of the monad as

\[
\text{type } \alpha \text{ mon } = \text{V of } \alpha \mid \text{E of } \text{exn list}
\]

\[
\text{let } \text{raise } e = \text{E [e]}
\]

\[
\text{let } \text{bind } m f =
\]

\[
\text{match } m \text{ with}
\]

\[
| \text{V } x \rightarrow f \ x
\]

\[
| \text{E exnlist } \rightarrow \text{E exnlist}
\]

For \(\text{bind } m f\), if \(m\) raises an exception, we have no value to pass to \(f\), so we cannot collect the exceptions raised by \(f\). (Unavoidable!)
Monads with nonstandard application

Rather than defining \(<*\>\) in terms of \(\text{bind}\), which will not collect all exceptions, we provide a more useful definition:

\[
\text{let } <*> f x = \\
\text{match } f, x \text{ with} \\
| V \text{ vf}, V \text{ vx} -> V (\text{vf} \text{ vx}) \\
| V \text{ vf}, E \text{ ex} -> E \text{ ex} \\
| E \text{ ef}, V \text{ vx} -> E \text{ ef} \\
| E \text{ ef}, E \text{ ex} -> E (\text{ef} @ \text{ex}) \text{ (* list concatenation *)}
\]

Thus, \(\text{ret } (+) <*> \text{raise } A <*> \text{raise } B\) produces \(E [A; B]\), as desired.
Applicative structures without bind

For some monads, we would like to compute both

- **static information** on the structure of the computation, and
- **a dynamic interpretation** that actually performs the computation, perhaps using the static information to be more efficient.

Simple example: for arithmetic expressions with variables, compute the free variables as static info and the function \( \text{environment} \rightarrow \text{value} \) as dynamic interpretation.

More realistic example: for parsing combinators, compute **nullable** and **first** information on the parsers as static info, and the function \( \text{text} \rightarrow \text{result} \times \text{text} \) as dynamic interpretation; use the static info to quickly eliminate impossible cases in the **either** combinator.
Free variables in the Environment monad

Let us try to extend the Environment monad with the computation of free variables.

\[
\text{type } \alpha \text{ mon } = \text{stringset} \times (\text{env} \to \alpha)
\]

(static information $\times$ dynamic interpretation)

let ret x = (emptyset, fun e -> x)

let getenv v = (singleton v, fun e -> map_lookup v e)

let <*> (sf, df) (sx, dx) =
  (union sf sx, fun e -> df (dx e) e)
Free variables in the Environment monad

However, bind cannot be defined!

```
type α mon = stringset * (env -> α)

let bind (sx,dx : α mon) (f: α -> β mon) : β mon =
  (union sx (fst (f ???)), fun e -> snd (f (dx e)))
```

When computing the static part of the result, we do not have any value to pass to function \( f \) so that we can extract the static part of \( f \)'s result!

(Besides: the static part of \( f \)'s result can depend on the value being passed to \( f \).)

→ This extended Environment is an applicative structure that is not a monad.
Composing applicative structures

Another evidence that applicative structures differ from monads is that applicative structures compose naturally: given

\[
\begin{align*}
\text{type } \alpha \text{ app1} & \quad \text{type } \alpha \text{ app2} \\
\text{pure1 : } \alpha \to \alpha \text{ app1} & \quad \text{pure2 : } \alpha \to \alpha \text{ app2} \\
<*>1 : (\alpha \to \beta) \to & \quad <*>2 : (\alpha \to \beta) \to \\
\alpha \text{ app1} \to \beta \text{ app1} & \quad \alpha \text{ app2} \to \beta \text{ app2}
\end{align*}
\]

we can define

\[
\begin{align*}
\text{type } \alpha \text{ app} & = \alpha \text{ app1 app2} \\
\text{let pure } (x : \alpha) : \alpha \text{ app} & = \text{pure2 (pure1 } x) \\
\text{let <*> } (f : \alpha \to \beta) : \alpha \text{ app} \to \beta \text{ app} & = <*>2 ( <*>1 f)
\end{align*}
\]
Outline

1. Introduction to monads
2. The monadic translation
   - Definition
   - Correctness
   - Application to some monads
3. Monadic programming
   - More examples of monads
   - Monad transformers
4. Bonus track: applicative structures
5. Bonus track: comonads
Comonads

The categorical dual of monads (what else?)

A comonad is defined by a parameterized type $\alpha \text{ com}$ and operations $\text{proj}$ and $\text{cobind}$, with types:

\[
\text{proj} : \forall \alpha. \alpha \text{ com} \to \alpha
\]
\[
\text{cobind} : \forall \alpha, \beta. (\alpha \text{ com} \to \beta) \to \alpha \text{ com} \to \beta \text{ com}
\]

The type $\tau \text{ com}$ is the type of processes that produce values of type $\tau$. (For example: a collection of $\tau$ values.)

$\text{proj } a$ extracts a value from such a process $a$.

$\text{cobind } f \ a$, given
– a function $f$ that produces a $\beta$ value from a $\alpha \text{ com}$ process,
– and a $\alpha \text{ com}$ process,
extends function $f$ to construct a process producing $\beta$’s.
Comonad laws

The \texttt{proj} and \texttt{cobind} operations of the comonad are supposed to satisfy the following algebraic laws:

\[
\begin{align*}
\text{proj}(\text{cobind } k x) & \approx k x \\
\text{cobind } \text{proj } x & \approx x \\
\text{cobind}(k_2 \circ \text{cobind } k_1) & \approx \text{cobind } k_2 \circ \text{cobind } k_1
\end{align*}
\]
Lazy evaluation as a comonad

module Lazy = struct

    type α com = α status ref
    and α status =
        | Evaluated of α
        | Suspended of unit -> α

    let proj (x: α com): α =
        match !x with
        | Evaluated v -> v
        | Suspended f -> let v = f() in x := Evaluated v; v

    let cobind (f: α com -> β) (x: α com): β com =
        ref (Suspended (fun () -> f x))

end
Lazy evaluation as a comonad

We can also equip lazy evaluation with monadic \texttt{ret} and monadic \texttt{bind}:

\begin{verbatim}
let ret (x: \alpha) : \alpha \text{ com} = ref (Evaluated x)

let bind (x: \alpha \text{ com}) (f: \alpha \rightarrow \beta \text{ com}): \beta \text{ com} = f (proj x)
\end{verbatim}

However, if we only have \texttt{ret} and \texttt{bind}, there is no way to suspend the evaluation of a nontrivial computation: \texttt{ret} always evaluates its argument!

In contrast, \texttt{cobind} \(f \ x\) is equivalent to \texttt{lazy}(f \ x) in Caml and let us suspend arbitrary computations.
Other uses of comonads

Working with infinite, lazy data structures: streams, bidirectional streams, etc.
(See example with cellular automata in companion file monads.ml)

Semantics of dataflow languages and reactive languages.